

1. ГРАММАТИКА ЯЗЫКА L

Здесь описана грамматика языка L. Степень формальности и строгости немного опущена для простоты изложения. За деталями, такими как множество бинарных операторов или определение идентификатора, следует обратиться к формальному описанию грамматики в файле `src/main/antlr/L.g4`.

1.1. Объявления функций. Тремя основными понятиями в грамматике языка L являются объявление функции (function definition), инструкция (statement) и выражение (expression).

Программа на языке L представляет собой набор из одного или нескольких объявлений функций, разделенных точкой с запятой.

```
<program> : <function definition> ( ';' <function definition> ) *  
<function definition> : <function name> '(' ( <arg name> ( ',' <arg name> ) * ) ? ')' <block>  
<block> : '{' <statement> ( ';' <statement> ) * '}'
```

Замечание: подразумевается, что точкой входа в программу является функция с именем **main**; если такой функции нет, то программа считается неисполняемой.

Замечание: здесь и далее всякий идентификатор, например, название функции или переменной, может совпадать с одним из ключевых слов языка; в случае неоднозначностей (например, если была объявлена функция с именем **write** или **read**) считается, что используется ключевое слово.

1.2. Инструкции. Из описания выше видно, что тело каждой функции представляет собой одну или несколько инструкций, разделенных точкой с запятой и обрамленных фигурными скобками. Инструкция может быть одной из следующих:

- Выражение
- Присваивание
- Возвращение значения из функции
- Вывод в выходной поток
- Чтение из входного потока
- Условное ветвление
- Цикл "пока"

```
<statement> : <expression>  
            | <variable name> ':=' <expression>  
            | 'return' <expression>  
            | 'write' '(' <expression> ')'  
            | 'read' '(' <variable name> ')'  
            | 'if' '(' <expression> ')' <block> ('else' <block>)?  
            | 'while' '(' <expression> ')' <block>
```

1.3. Выражения. Выражение может быть одним из следующих:

- Выражение, взятое в скобки
- Взятие значения переменной
- Вызов функции
- Целое число
- Число с плавающей точкой
- Применение бинарной операции к двум выражениям

```
<expression> : '(' <expression> ')'  
             | <variable name>  
             | <function name> '(' ( <expression> ( ',' <expression> ) * ) ? ')'  
             | <decimal integer literal>  
             | <decimal floating point literal>  
             | '(' <expression> <binary operation> <expression> ')'
```

Замечание: в L нет булевых литералов, вместо них используются целые числа, как в языках семейства C: ноль означает ложь, а любое другое число — истину.